// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.2) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: address zero is not a valid owner");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _ownerOf(tokenId);
        require(owner != address(0), "ERC721: invalid token ID");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        _requireMinted(tokenId);

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overridden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not token owner or approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        _requireMinted(tokenId);

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner or approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner or approved");
        _safeTransfer(from, to, tokenId, data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns the owner of the `tokenId`. Does NOT revert if token doesn't exist
     */
    function _ownerOf(uint256 tokenId) internal view virtual returns (address) {
        return _owners[tokenId];
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _ownerOf(tokenId) != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || isApprovedForAll(owner, spender) || getApproved(tokenId) == spender);
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId, 1);

        // Check that tokenId was not minted by `_beforeTokenTransfer` hook
        require(!_exists(tokenId), "ERC721: token already minted");

        unchecked {
            // Will not overflow unless all 2**256 token ids are minted to the same owner.
            // Given that tokens are minted one by one, it is impossible in practice that
            // this ever happens. Might change if we allow batch minting.
            // The ERC fails to describe this case.
            _balances[to] += 1;
        }

        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId, 1);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     * This is an internal function that does not check if the sender is authorized to operate on the token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId, 1);

        // Update ownership in case tokenId was transferred by `_beforeTokenTransfer` hook
        owner = ERC721.ownerOf(tokenId);

        // Clear approvals
        delete _tokenApprovals[tokenId];

        unchecked {
            // Cannot overflow, as that would require more tokens to be burned/transferred
            // out than the owner initially received through minting and transferring in.
            _balances[owner] -= 1;
        }
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId, 1);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId, 1);

        // Check that tokenId was not transferred by `_beforeTokenTransfer` hook
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");

        // Clear approvals from the previous owner
        delete _tokenApprovals[tokenId];

        unchecked {
            // `_balances[from]` cannot overflow for the same reason as described in `_burn`:
            // `from`'s balance is the number of token held, which is at least one before the current
            // transfer.
            // `_balances[to]` could overflow in the conditions described in `_mint`. That would require
            // all 2**256 token ids to be minted, which in practice is impossible.
            _balances[from] -= 1;
            _balances[to] += 1;
        }
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId, 1);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits an {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits an {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Reverts if the `tokenId` has not been minted yet.
     */
    function _requireMinted(uint256 tokenId) internal view virtual {
        require(_exists(tokenId), "ERC721: invalid token ID");
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    /// @solidity memory-safe-assembly
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting and burning. If {ERC721Consecutive} is
     * used, the hook may be called as part of a consecutive (batch) mint, as indicated by `batchSize` greater than 1.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s tokens will be transferred to `to`.
     * - When `from` is zero, the tokens will be minted for `to`.
     * - When `to` is zero, ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     * - `batchSize` is non-zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 firstTokenId,
        uint256 batchSize
    ) internal virtual {}

    /**
     * @dev Hook that is called after any token transfer. This includes minting and burning. If {ERC721Consecutive} is
     * used, the hook may be called as part of a consecutive (batch) mint, as indicated by `batchSize` greater than 1.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s tokens were transferred to `to`.
     * - When `from` is zero, the tokens were minted for `to`.
     * - When `to` is zero, ``from``'s tokens were burned.
     * - `from` and `to` are never both zero.
     * - `batchSize` is non-zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 firstTokenId,
        uint256 batchSize
    ) internal virtual {}

    /**
     * @dev Unsafe write access to the balances, used by extensions that "mint" tokens using an {ownerOf} override.
     *
     * WARNING: Anyone calling this MUST ensure that the balances remain consistent with the ownership. The invariant
     * being that for any address `a` the value returned by `balanceOf(a)` must be equal to the number of tokens such
     * that `ownerOf(tokenId)` is `a`.
     */
    // solhint-disable-next-line func-name-mixedcase
    function __unsafe_increaseBalance(address account, uint256 amount) internal {
        _balances[account] += amount;
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/ERC721Enumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/extensions/ERC721Enumerable.sol)

pragma solidity ^0.8.0;

import "../ERC721.sol";
import "./IERC721Enumerable.sol";

/**
 * @dev This implements an optional extension of {ERC721} defined in the EIP that adds
 * enumerability of all the token ids in the contract as well as all token ids owned by each
 * account.
 */
abstract contract ERC721Enumerable is ERC721, IERC721Enumerable {
    // Mapping from owner to list of owned token IDs
    mapping(address => mapping(uint256 => uint256)) private _ownedTokens;

    // Mapping from token ID to index of the owner tokens list
    mapping(uint256 => uint256) private _ownedTokensIndex;

    // Array with all token ids, used for enumeration
    uint256[] private _allTokens;

    // Mapping from token id to position in the allTokens array
    mapping(uint256 => uint256) private _allTokensIndex;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC721) returns (bool) {
        return interfaceId == type(IERC721Enumerable).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721Enumerable-tokenOfOwnerByIndex}.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721.balanceOf(owner), "ERC721Enumerable: owner index out of bounds");
        return _ownedTokens[owner][index];
    }

    /**
     * @dev See {IERC721Enumerable-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _allTokens.length;
    }

    /**
     * @dev See {IERC721Enumerable-tokenByIndex}.
     */
    function tokenByIndex(uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721Enumerable.totalSupply(), "ERC721Enumerable: global index out of bounds");
        return _allTokens[index];
    }

    /**
     * @dev See {ERC721-_beforeTokenTransfer}.
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 firstTokenId,
        uint256 batchSize
    ) internal virtual override {
        super._beforeTokenTransfer(from, to, firstTokenId, batchSize);

        if (batchSize > 1) {
            // Will only trigger during construction. Batch transferring (minting) is not available afterwards.
            revert("ERC721Enumerable: consecutive transfers not supported");
        }

        uint256 tokenId = firstTokenId;

        if (from == address(0)) {
            _addTokenToAllTokensEnumeration(tokenId);
        } else if (from != to) {
            _removeTokenFromOwnerEnumeration(from, tokenId);
        }
        if (to == address(0)) {
            _removeTokenFromAllTokensEnumeration(tokenId);
        } else if (to != from) {
            _addTokenToOwnerEnumeration(to, tokenId);
        }
    }

    /**
     * @dev Private function to add a token to this extension's ownership-tracking data structures.
     * @param to address representing the new owner of the given token ID
     * @param tokenId uint256 ID of the token to be added to the tokens list of the given address
     */
    function _addTokenToOwnerEnumeration(address to, uint256 tokenId) private {
        uint256 length = ERC721.balanceOf(to);
        _ownedTokens[to][length] = tokenId;
        _ownedTokensIndex[tokenId] = length;
    }

    /**
     * @dev Private function to add a token to this extension's token tracking data structures.
     * @param tokenId uint256 ID of the token to be added to the tokens list
     */
    function _addTokenToAllTokensEnumeration(uint256 tokenId) private {
        _allTokensIndex[tokenId] = _allTokens.length;
        _allTokens.push(tokenId);
    }

    /**
     * @dev Private function to remove a token from this extension's ownership-tracking data structures. Note that
     * while the token is not assigned a new owner, the `_ownedTokensIndex` mapping is _not_ updated: this allows for
     * gas optimizations e.g. when performing a transfer operation (avoiding double writes).
     * This has O(1) time complexity, but alters the order of the _ownedTokens array.
     * @param from address representing the previous owner of the given token ID
     * @param tokenId uint256 ID of the token to be removed from the tokens list of the given address
     */
    function _removeTokenFromOwnerEnumeration(address from, uint256 tokenId) private {
        // To prevent a gap in from's tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = ERC721.balanceOf(from) - 1;
        uint256 tokenIndex = _ownedTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary
        if (tokenIndex != lastTokenIndex) {
            uint256 lastTokenId = _ownedTokens[from][lastTokenIndex];

            _ownedTokens[from][tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
            _ownedTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index
        }

        // This also deletes the contents at the last position of the array
        delete _ownedTokensIndex[tokenId];
        delete _ownedTokens[from][lastTokenIndex];
    }

    /**
     * @dev Private function to remove a token from this extension's token tracking data structures.
     * This has O(1) time complexity, but alters the order of the _allTokens array.
     * @param tokenId uint256 ID of the token to be removed from the tokens list
     */
    function _removeTokenFromAllTokensEnumeration(uint256 tokenId) private {
        // To prevent a gap in the tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = _allTokens.length - 1;
        uint256 tokenIndex = _allTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary. However, since this occurs so
        // rarely (when the last minted token is burnt) that we still do the swap here to avoid the gas cost of adding
        // an 'if' statement (like in _removeTokenFromOwnerEnumeration)
        uint256 lastTokenId = _allTokens[lastTokenIndex];

        _allTokens[tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
        _allTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index

        // This also deletes the contents at the last position of the array
        delete _allTokensIndex[tokenId];
        _allTokens.pop();
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC721/extensions/IERC721Enumerable.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {
    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result * 8) < value ? 1 : 0);
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: contracts/Hexamillennia.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.18;

import '@openzeppelin/contracts/token/ERC721/extensions/ERC721Enumerable.sol';
import '@openzeppelin/contracts/access/Ownable.sol';
import './HexamillenniaAlgorithm.sol';

contract Hexamillennia is ERC721Enumerable, Ownable {
    uint256 public constant MAX_SUPPLY = 1000;

    bool public active;
    mapping(uint256 => uint256) public randomSource;

    constructor() ERC721('Hexamillennia', 'HXMLLNN') {}

    function activate() external onlyOwner {
        active = true;
    }

    function mintTiling() external {
        require(active, 'Mint not active');
        uint256 tokenId = totalSupply();
        require(tokenId < MAX_SUPPLY, 'Max supply reached');
        randomSource[tokenId] = uint256(keccak256(abi.encodePacked(msg.sender, blockhash(block.number - 1), tokenId)));
        _mint(msg.sender, tokenId);
    }

    function tokenURI(uint256 tokenId) public view override returns (string memory) {
        _requireMinted(tokenId);
        return HexamillenniaAlgorithm.tokenURI(tokenId, randomSource[tokenId]);
    }

    function tokenSVG(uint256 tokenId) public view returns (string memory) {
        _requireMinted(tokenId);
        return HexamillenniaAlgorithm.tokenSVG(tokenId, randomSource[tokenId]);
    }
}

// File: contracts/HexamillenniaAlgorithm.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.18;

uint256 constant MASK_4 = 2 ** 4 - 1;
uint256 constant MASK_6 = 2 ** 6 - 1;
uint256 constant MASK_8 = 2 ** 8 - 1;
uint256 constant MASK_12 = 2 ** 12 - 1;
uint256 constant MASK_16 = 2 ** 16 - 1;
uint256 constant MASK_32 = 2 ** 32 - 1;

uint256 constant HEXAGON_PERP_WALK_MASK = 2 ** 6;
uint256 constant HEXAGON_PERP_2_WALK_MASK = 2 ** 8;
uint256 constant HEXAGON_PERP_3_WALK_MASK = 2 ** 9;

uint256 constant HEXAGON_PAINT_OFFSET = 12;
uint256 constant SQUARE_PAINT_OFFSET = 16;
uint256 constant SQUARE_2_PAINT_OFFSET = 24;
uint256 constant TRIANGLE_PAINT_OFFSET = 28;
uint256 constant TRIANGLE_1_PAINT_OFFSET = 32;
uint256 constant HEXAGON_PAINT_MASK = 0xf000;
uint256 constant SQUARE_PAINT_MASK = 0xf0000;
uint256 constant TRIANGLE_PAINT_MASK = 0xf0000000;
uint256 constant TRIANGLE_1_PAINT_MASK = 0xf00000000;
uint256 constant SQUARE_2_WALK_PAINT_MASK = 0xf000100;

uint256 constant HEXAGON_EXPAND_OFFSET = 36;
uint256 constant HEXAGON_PERP_EXPAND_OFFSET = 42;
uint256 constant HEXAGON_EXPAND_MASK = 0x1000000000;
uint256 constant HEXAGON_PERP_EXPAND_MASK = 0x40000000000;
uint256 constant HEXAGON_FULL_BOUNDARY_MASK = 0x3f000000000;
uint256 constant SQUARE_HALF_BOUNDARY_MASK = 0x41000000000;
uint256 constant VERTEX_013_EXPAND_MASK = 0x82143021810c086043;

uint256 constant EXPAND_ROOT_OFFSET = 48;
uint256 constant SELF_OFFSET = 52;
uint256 constant ROW_COL_OFFSET = 68;
uint256 constant STATE_OFFSET = 84;
uint256 constant STATE_1_OFFSET = 100;
uint256 constant STATE_3_OFFSET = 132;
uint256 constant STATE_5_OFFSET = 164;

uint256 constant ANGLE_TO_HEXAGON = 0x0100010100017f007f7f007f;
uint256 constant ANGLE_TO_HEXAGON_PLUS = 0x020102020102000100000100;
uint256 constant ANGLE_TO_VERTEX_X = 0x000003e8000007d0000003e8fffffc18fffff830fffffc18;
uint256 constant ANGLE_TO_VERTEX_Y = 0xfffff93c00000000000006c4000006c400000000fffff93c;

uint256 constant POPCOUNT_6 = 0x6554544354434332544343324332322154434332433232214332322132212110;

// These memory locations are above the area that Solidity allocates for the 3 string constants below, and this is the only area that Solidity
// allocates when this library is used as intended.
uint256 constant LOG2_DIM_M = 0x680;
uint256 constant DIM_M = 0x6a0;
uint256 constant UNROLLED_GRID_M = 0x6c0;
uint256 constant OPEN_M = 0x6e0;
uint256 constant MARGIN_M = 0x700;
uint256 constant UNROLLED_GRID_ROWS_M = 0x720;
uint256 constant UNROLLED_GRID_COLS_M = 0x740;
uint256 constant STACK_M = 0x760;
uint256 constant STACK_IDX_M = 0x780;
uint256 constant CACHE_M = 0x7a0;
uint256 constant CACHE_IDX_M = 0x7c0;
uint256 constant OUTPUT_M = 0x7e0;
uint256 constant OUTPUT_IDX_M = 0x800;
uint256 constant STATE_M_M = 0x820;
uint256 constant ANGLE_M = 0x840;
uint256 constant STEPS_IDX_M = 0x860;
uint256 constant STEPS_M = 0x880;
uint256 constant PALETTE_IDX_M = 0x8a0;
uint256 constant PALETTE_M = 0x8c0;
uint256 constant NUM_COLORS_M = 0x8e0;
uint256 constant COLOR_M = 0x900;
uint256 constant EDGE_COUNT_M = 0x920;
uint256 constant SVG_STRING_LOOKUP_M = 0x940;
uint256 constant OPEN_VIEW_BOX_X_DECIMAL_M = 0x960;
uint256 constant OPEN_VIEW_BOX_X_DECIMAL_LENGTH_M = 0x980;
uint256 constant OPEN_VIEW_BOX_Y_DECIMAL_M = 0x9a0;
uint256 constant OPEN_VIEW_BOX_Y_DECIMAL_LENGTH_M = 0x9c0;
uint256 constant OPEN_VIEW_BOX_WIDTH_DECIMAL_M = 0x9e0;
uint256 constant OPEN_VIEW_BOX_WIDTH_DECIMAL_LENGTH_M = 0xa00;
uint256 constant OPEN_VIEW_BOX_HEIGHT_DECIMAL_M = 0xa20;
uint256 constant OPEN_VIEW_BOX_HEIGHT_DECIMAL_LENGTH_M = 0xa40;
uint256 constant DOMAIN_WIDTH_DECIMAL_M = 0xa60;
uint256 constant DOMAIN_WIDTH_DECIMAL_LENGTH_M = 0xa80;
uint256 constant DOMAIN_HEIGHT_DECIMAL_M = 0xaa0;
uint256 constant DOMAIN_HEIGHT_DECIMAL_LENGTH_M = 0xac0;
uint256 constant SVG_START_M = 0xae0;
uint256 constant SVG_END_M = 0xb00;
uint256 constant JSON_STRING_LOOKUP_M = 0xb20;
uint256 constant TOKEN_ID_M = 0xb40;
uint256 constant TOKEN_ID_DECIMAL_M = 0xb60;
uint256 constant TOKEN_ID_DECIMAL_LENGTH_M = 0xb80;
uint256 constant DIM_DECIMAL_M = 0xba0;
uint256 constant DIM_DECIMAL_LENGTH_M = 0xbc0;
uint256 constant PALETTE_IDX_DECIMAL_M = 0xbe0;
uint256 constant PALETTE_IDX_DECIMAL_LENGTH_M = 0xc00;

uint256 constant ANGLE_EDGE_TO_VECTOR = 0xcc0;
uint256 constant ANGLE_EDGE_TO_VECTOR_OFFSET = 0xece4dcd1c7bbb0a69b90867f7770695e52483d31261b0f0700;
uint256 constant BASE64 = 0xda1;
uint256 constant RANDOM_SOURCE = 0xe00;
uint256 constant SHIFT_M = 0xe20;
uint256 constant GRID = 0xe40;

string constant PALETTES = 'FF87CA7FAEFAB07676FCDED4F7ABD4CCA3A3B8D1FFA555ECC47AFFFDFF00FFF8BC38E54DCFFF8DFF731DF7A76CE5DFD6FBF8F4807E7D633E35A27B5C9FC088F4DFBAFFFEA9379237BA3A33D85C2BF1F582FB6B337D3D443F2828FEE1830280C0253978B6E6FFD3E0EFA1F7FF84A1C9EDF5FC540375FF7000FF4949FFFD8C824C96F2D0A3D0A369B46F37A70A0D800004BF040AB51212EFEFEFDFDEDEFFFFFFFEFF9F393E465D697AF3CCFFF9DEFCD3B5F5F6EBFA012106210101200D073412115F3D36F4E1BCE3C69DB5918852230E864123F2BD77';
uint256 constant PALETTES_OFFSET = 0x473e3835312b261f18130d0700;
uint256 constant NUM_PALETTES = 12;
string constant SVG_STRING_LOOKUP = '<svg xmlns="http://www.w3.org/2000/svg" viewBox=""><rect x="-2732" y="-2732" width="" height="" fill="white"/><g stroke="black" stroke-width="100" stroke-linejoin="round" stroke-linecap="round" fill-rule="evenodd"><path d="" fill="#"/></g></svg> 0 0 -2732 -2732 M-2732 -2732l0 ';
string constant JSON_STRING_LOOKUP = 'data:application/json,%7B%22name%22:%22Tiling%20%22,%22description%22:%22Hexamillennia%20is%20generated%20entirely%20on%20the%20EVM.%20Released%20under%20CC0.%22,%22attributes%22:%5B%7B%22trait_type%22:%22%22,%22value%22:%22%22%7D,%7B%22trait_type%22:%22%22,%22value%22:%22%22%7D%5D,%22image%22:%22data:image/svg+xml;base64,%22%7DSizeFormStepsPaletteClosedOpenLowMediumHigh';

library HexamillenniaAlgorithm {
    function tokenURI(uint256 tokenId, uint256 randomSource) internal pure returns (string memory) {
        generateSVG(tokenId, randomSource);
        resetOutput();
        writeJSON();
        returnOutput();
    }

    function tokenSVG(uint256 tokenId, uint256 randomSource) internal pure returns (string memory) {
        generateSVG(tokenId, randomSource);
        returnOutput();
    }

    function generateSVG(uint256 tokenId, uint256 randomSource) internal pure {
        initializeKnownData(tokenId, randomSource);
        chooseAttributes();
        initializeVariables();
        prepareGrid();
        walk();
        paint();
        prepareUnrolledGrid();
        writeDecimalLookup();
        resetOutput();
        writePreExpand();
        expand();
        writePostExpand();
    }

    function initializeKnownData(uint256 tokenId, uint256 randomSource) internal pure {
        string memory palettes = PALETTES;
        string memory svgStringLookup = SVG_STRING_LOOKUP;
        string memory jsonStringLookup = JSON_STRING_LOOKUP;
        assembly {
            mstore(TOKEN_ID_M, tokenId)
            mstore(RANDOM_SOURCE, randomSource)
            mstore(PALETTE_M, add(palettes, 0x20))
            mstore(SVG_STRING_LOOKUP_M, add(svgStringLookup, 0x20))
            mstore(JSON_STRING_LOOKUP_M, add(jsonStringLookup, 0x20))
            mstore(0xcc0, ' 2000 0 0 -2000 -1732 -1000 -100')
            mstore(0xce0, '0 1732 1000 -1732 -1732 -1000 -1')
            mstore(0xd00, '732 1000 1000 1732 -1000 -1732 -')
            mstore(0xd20, '1732 1000 0 2000 2000 0 -2000 0 ')
            mstore(0xd40, '0 2000 1732 1000 1000 -1732 -100')
            mstore(0xd60, '0 1732 1732 1000 1732 -1000 -100')
            mstore(0xd80, '0 -1732 1000 1732 1732 -1000 0 -')
            mstore(0xda0, '2000 -2000 0')
            mstore(0xdc0, 'ABCDEFGHIJKLMNOPQRSTUVWXYZabcdef')
            mstore(0xde0, 'ghijklmnopqrstuvwxyz0123456789+/')
        }
    }

    function chooseAttributes() internal pure {
        assembly {
            function updateRandomSource() {
                let shift := mload(SHIFT_M)
                shift := add(shift, 8)
                if eq(shift, 256) {
                    mstore(RANDOM_SOURCE, keccak256(RANDOM_SOURCE, 0x20))
                    shift := 0
                }
                mstore(SHIFT_M, shift)
            }
            // log2Dim can be at most 4 in this implementation
            let log2Dim := add(shr(6, and(shr(mload(SHIFT_M), mload(RANDOM_SOURCE)), MASK_8)), 1)
            let dim := shl(log2Dim, 1)
            mstore(LOG2_DIM_M, log2Dim)
            mstore(DIM_M, dim)
            updateRandomSource()
            mstore(OPEN_M, shr(7, and(shr(mload(SHIFT_M), mload(RANDOM_SOURCE)), MASK_8)))
            updateRandomSource()
            mstore(STATE_M_M, add(GRID, shl(5, shr(8, mul(mul(dim, dim), and(shr(mload(SHIFT_M), mload(RANDOM_SOURCE)), MASK_8))))))
            updateRandomSource()
            mstore(ANGLE_M, shr(8, mul(6, and(shr(mload(SHIFT_M), mload(RANDOM_SOURCE)), MASK_8))))
            updateRandomSource()
            let stepsIdx := shr(8, mul(3, and(shr(mload(SHIFT_M), mload(RANDOM_SOURCE)), MASK_8)))
            mstore(STEPS_IDX_M, stepsIdx)
            mstore(STEPS_M, shl(add(add(stepsIdx, 4), shl(1, log2Dim)), 1))
            updateRandomSource()
            let paletteIdx := shr(8, mul(NUM_PALETTES, and(shr(mload(SHIFT_M), mload(RANDOM_SOURCE)), MASK_8)))
            mstore(PALETTE_IDX_M, paletteIdx)
            let adjusted := shr(shl(3, paletteIdx), PALETTES_OFFSET)
            let offset := and(adjusted, MASK_8)
            mstore(PALETTE_M, add(mload(PALETTE_M), mul(offset, 0x6)))
            mstore(NUM_COLORS_M, sub(and(shr(8, adjusted), MASK_8), offset))
            updateRandomSource()
        }
    }

    function initializeVariables() internal pure {
        assembly {
            let dim := mload(DIM_M)
            let open := mload(OPEN_M)
            let margin := mul(sub(mload(LOG2_DIM_M), 1), open)
            let hexagonCount
            switch open
            case 0 {
                hexagonCount := mul(shr(1, add(dim, 2)), add(shl(1, dim), 3))
            }
            case 1 {
                hexagonCount := mul(dim, dim)
            }
            mstore(UNROLLED_GRID_M, add(GRID, shl(5, mul(dim, dim))))
            mstore(MARGIN_M, margin)
            mstore(UNROLLED_GRID_ROWS_M, add(add(add(dim, shr(1, dim)), 4), shl(1, margin)))
            mstore(UNROLLED_GRID_COLS_M, add(add(dim, 4), shl(1, margin)))
            mstore(STACK_M, add(mload(UNROLLED_GRID_M), shl(5, mul(mload(UNROLLED_GRID_ROWS_M), mload(UNROLLED_GRID_COLS_M)))))
            mstore(STACK_IDX_M, mload(STACK_M))
            mstore(CACHE_M, add(mload(STACK_M), add(mul(hexagonCount, 18), 0x20)))
            mstore(CACHE_IDX_M, mload(CACHE_M))
            mstore(OUTPUT_M, add(mload(CACHE_M), add(mul(hexagonCount, 39), 0x20)))
            mstore(OUTPUT_IDX_M, mload(OUTPUT_M))
        }
    }

    function prepareGrid() internal pure {
        assembly {
            let log2Dim := mload(LOG2_DIM_M)
            let dim := mload(DIM_M)
            let gridCount := mul(dim, dim)
            for {
                let gridIdx
            } lt(gridIdx, gridCount) {
                gridIdx := add(gridIdx, 1)
            } {
                let stateM := add(GRID, shl(5, gridIdx))
                mstore(stateM, shl(SELF_OFFSET, stateM))
                let row := shr(log2Dim, gridIdx)
                let col := and(gridIdx, sub(dim, 1))
                switch and(and(and(gt(row, 0), lt(row, sub(dim, 1))), gt(col, 0)), lt(col, sub(dim, 1)))
                case 0 {
                    for {
                        let angle
                    } lt(angle, 6) {
                        angle := add(angle, 1)
                    } {
                        let hexagonR0 := add(or(shl(8, col), row), shr(shl(4, angle), ANGLE_TO_HEXAGON))
                        let col0 := shr(8, hexagonR0)
                        mstore(
                            stateM,
                            or(
                                mload(stateM),
                                shl(
                                    add(shl(4, angle), STATE_OFFSET),
                                    add(
                                        add(GRID, shl(add(log2Dim, 5), and(sub(hexagonR0, mul(shr(log2Dim, col0), shr(1, dim))), sub(dim, 1)))),
                                        shl(5, and(col0, sub(dim, 1)))
                                    )
                                )
                            )
                        )
                    }
                }
                case 1 {
                    mstore(
                        stateM,
                        or(
                            mload(stateM),
                            shl(
                                STATE_OFFSET,
                                or(
                                    or(
                                        or(
                                            or(
                                                or(
                                                    shl(80, add(add(GRID, shl(add(log2Dim, 5), row)), shl(5, add(col, 1)))),
                                                    shl(64, add(add(GRID, shl(add(log2Dim, 5), add(row, 1))), shl(5, add(col, 1))))
                                                ),
                                                shl(48, add(add(GRID, shl(add(log2Dim, 5), add(row, 1))), shl(5, col)))
                                            ),
                                            shl(32, add(add(GRID, shl(add(log2Dim, 5), row)), shl(5, sub(col, 1))))
                                        ),
                                        shl(16, add(add(GRID, shl(add(log2Dim, 5), sub(row, 1))), shl(5, sub(col, 1))))
                                    ),
                                    add(add(GRID, shl(add(log2Dim, 5), sub(row, 1))), shl(5, col))
                                )
                            )
                        )
                    )
                }
            }
        }
    }

    function walk() internal pure {
        assembly {
            let stateM := mload(STATE_M_M)
            let angle := mload(ANGLE_M)
            let steps := mload(STEPS_M)
            for {
                let i
            } lt(i, steps) {
                i := add(i, 1)
            } {
                let shift := mload(add(RANDOM_SOURCE, 0x20))
                switch shr(6, and(shr(shift, mload(RANDOM_SOURCE)), MASK_8))
                case 0 {
                    mstore(stateM, or(mload(stateM), shl(angle, 1)))
                    angle := addmod(angle, 5, 6)
                }
                case 1 {
                    mstore(stateM, or(mload(stateM), shl(angle, HEXAGON_PERP_WALK_MASK)))
                    stateM := and(shr(add(shl(4, angle), STATE_OFFSET), mload(stateM)), MASK_16)
                    angle := addmod(angle, 2, 6)
                }
                case 2 {
                    stateM := and(shr(add(shl(4, addmod(angle, 1, 6)), STATE_OFFSET), mload(stateM)), MASK_16)
                    angle := addmod(angle, 4, 6)
                    mstore(stateM, or(mload(stateM), shl(angle, HEXAGON_PERP_WALK_MASK)))
                }
                case 3 {
                    angle := addmod(angle, 1, 6)
                    mstore(stateM, or(mload(stateM), shl(angle, 1)))
                }
                if eq(shift, 248) {
                    mstore(RANDOM_SOURCE, keccak256(RANDOM_SOURCE, 0x20))
                    mstore(SHIFT_M, 0)
                    continue
                }
                mstore(SHIFT_M, add(shift, 8))
            }
        }
    }

    function paint() internal pure {
        assembly {
            function paintDF() {
                for {

                } gt(mload(STACK_IDX_M), mload(STACK_M)) {

                } {
                    mstore(STACK_IDX_M, sub(mload(STACK_IDX_M), 0x3))
                    let top := shr(232, mload(mload(STACK_IDX_M)))
                    switch shr(20, top)
                    case 0 {
                        for {
                            let angle
                        } lt(angle, 3) {
                            angle := add(angle, 1)
                        } {
                            if iszero(and(mload(top), or(shl(angle, 1), shl(shl(2, angle), SQUARE_PAINT_MASK)))) {
                                mstore(top, or(mload(top), shl(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, angle)), top)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                            let stateR3M := and(shr(add(shl(4, add(angle, 3)), STATE_OFFSET), mload(top)), MASK_16)
                            if iszero(or(and(mload(top), shl(add(angle, 3), 1)), and(mload(stateR3M), shl(shl(2, angle), SQUARE_PAINT_MASK)))) {
                                mstore(stateR3M, or(mload(stateR3M), shl(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, angle)), stateR3M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                    }
                    case 1 {
                        let angle := and(shr(16, top), MASK_4)
                        let stateM := and(top, MASK_16)
                        if iszero(and(mload(stateM), or(shl(angle, 1), HEXAGON_PAINT_MASK))) {
                            mstore(stateM, or(mload(stateM), shl(HEXAGON_PAINT_OFFSET, mload(COLOR_M))))
                            mstore(mload(STACK_IDX_M), shl(232, stateM))
                            mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        }
                        let stateR0M := and(shr(add(shl(4, angle), STATE_OFFSET), mload(stateM)), MASK_16)
                        if iszero(and(mload(stateR0M), or(shl(add(angle, 3), 1), HEXAGON_PAINT_MASK))) {
                            mstore(stateR0M, or(mload(stateR0M), shl(HEXAGON_PAINT_OFFSET, mload(COLOR_M))))
                            mstore(mload(STACK_IDX_M), shl(232, stateR0M))
                            mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        }
                        switch eq(angle, 2)
                        case 0 {
                            if iszero(and(mload(stateM), or(shl(angle, HEXAGON_PERP_WALK_MASK), shl(shl(2, angle), TRIANGLE_PAINT_MASK)))) {
                                mstore(stateM, or(mload(stateM), shl(add(shl(2, angle), TRIANGLE_PAINT_OFFSET), mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(or(0x200000, shl(16, angle)), stateM)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        case 1 {
                            let state3M := and(shr(STATE_3_OFFSET, mload(stateM)), MASK_16)
                            if iszero(or(and(mload(stateM), HEXAGON_PERP_2_WALK_MASK), and(mload(state3M), TRIANGLE_PAINT_MASK))) {
                                mstore(state3M, or(mload(state3M), shl(TRIANGLE_PAINT_OFFSET, mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(0x200000, state3M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        switch eq(angle, 0)
                        case 0 {
                            if iszero(
                                or(
                                    and(mload(stateR0M), shl(add(angle, 3), HEXAGON_PERP_WALK_MASK)),
                                    and(mload(stateM), shl(shl(2, sub(angle, 1)), TRIANGLE_PAINT_MASK))
                                )
                            ) {
                                mstore(stateM, or(mload(stateM), shl(add(shl(2, sub(angle, 1)), TRIANGLE_PAINT_OFFSET), mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(or(0x200000, shl(16, sub(angle, 1))), stateM)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        case 1 {
                            let state5M := and(shr(STATE_5_OFFSET, mload(stateM)), MASK_16)
                            if iszero(or(and(mload(stateR0M), HEXAGON_PERP_3_WALK_MASK), and(mload(state5M), TRIANGLE_1_PAINT_MASK))) {
                                mstore(state5M, or(mload(state5M), shl(TRIANGLE_1_PAINT_OFFSET, mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(0x210000, state5M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                    }
                    case 2 {
                        let angle := and(shr(16, top), MASK_4)
                        let stateM := and(top, MASK_16)
                        let stateR1M := and(shr(add(shl(4, add(angle, 1)), STATE_OFFSET), mload(stateM)), MASK_16)
                        if iszero(and(mload(stateM), or(shl(angle, HEXAGON_PERP_WALK_MASK), shl(shl(2, angle), SQUARE_PAINT_MASK)))) {
                            mstore(stateM, or(mload(stateM), shl(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                            mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, angle)), stateM)))
                            mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        }
                        if iszero(
                            or(
                                and(mload(stateR1M), shl(add(angle, 4), HEXAGON_PERP_WALK_MASK)),
                                and(mload(stateM), shl(shl(2, add(angle, 1)), SQUARE_PAINT_MASK))
                            )
                        ) {
                            mstore(stateM, or(mload(stateM), shl(add(shl(2, add(angle, 1)), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                            mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, add(angle, 1))), stateM)))
                            mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        }
                        switch angle
                        case 0 {
                            let state0M := and(shr(STATE_OFFSET, mload(stateM)), MASK_16)
                            if iszero(and(mload(state0M), SQUARE_2_WALK_PAINT_MASK)) {
                                mstore(state0M, or(mload(state0M), shl(SQUARE_2_PAINT_OFFSET, mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(0x120000, state0M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        case 1 {
                            if iszero(
                                or(
                                    and(mload(and(shr(STATE_1_OFFSET, mload(stateM)), MASK_16)), HEXAGON_PERP_3_WALK_MASK),
                                    and(mload(stateR1M), SQUARE_PAINT_MASK)
                                )
                            ) {
                                mstore(stateR1M, or(mload(stateR1M), shl(SQUARE_PAINT_OFFSET, mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(0x100000, stateR1M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                    }
                }
            }
            function chooseColor() {
                let shift := mload(SHIFT_M)
                mstore(COLOR_M, add(shr(8, mul(mload(NUM_COLORS_M), and(shr(shift, mload(RANDOM_SOURCE)), MASK_8))), 1))
                shift := add(shift, 8)
                if eq(shift, 256) {
                    mstore(RANDOM_SOURCE, keccak256(RANDOM_SOURCE, 0x20))
                    shift := 0
                }
                mstore(SHIFT_M, shift)
            }
            let gridEnd := mload(UNROLLED_GRID_M)
            for {
                let stateM := GRID
            } lt(stateM, gridEnd) {
                stateM := add(stateM, 0x20)
            } {
                if iszero(and(mload(stateM), HEXAGON_PAINT_MASK)) {
                    chooseColor()
                    mstore(stateM, or(mload(stateM), shl(HEXAGON_PAINT_OFFSET, mload(COLOR_M))))
                    mstore(mload(STACK_IDX_M), shl(232, stateM))
                    mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                    paintDF()
                }
                for {
                    let angle
                } lt(angle, 3) {
                    angle := add(angle, 1)
                } {
                    if iszero(and(mload(stateM), shl(shl(2, angle), SQUARE_PAINT_MASK))) {
                        chooseColor()
                        mstore(stateM, or(mload(stateM), shl(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                        mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, angle)), stateM)))
                        mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        paintDF()
                    }
                    if and(lt(angle, 2), iszero(and(mload(stateM), shl(shl(2, angle), TRIANGLE_PAINT_MASK)))) {
                        chooseColor()
                        mstore(stateM, or(mload(stateM), shl(add(shl(2, angle), TRIANGLE_PAINT_OFFSET), mload(COLOR_M))))
                        mstore(mload(STACK_IDX_M), shl(232, or(or(0x200000, shl(16, angle)), stateM)))
                        mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        paintDF()
                    }
                }
            }
        }
    }

    function prepareUnrolledGrid() internal pure {
        assembly {
            let dim := mload(DIM_M)
            let unrolledGrid := mload(UNROLLED_GRID_M)
            let margin := mload(MARGIN_M)
            let cols := mload(UNROLLED_GRID_COLS_M)
            let gridCount := mul(mload(UNROLLED_GRID_ROWS_M), cols)
            for {
                let gridIdx
            } lt(gridIdx, gridCount) {
                gridIdx := add(gridIdx, 1)
            } {
                let stateM := add(unrolledGrid, shl(5, gridIdx))
                let row := div(gridIdx, cols)
                let col := mod(gridIdx, cols)
                if and(
                    and(and(gt(col, 0), lt(col, add(add(shl(1, margin), 3), dim))), gt(add(shl(1, row), 1), col)),
                    lt(shl(1, row), add(add(add(shl(1, margin), 3), shl(1, dim)), col))
                ) {
                    let colN := sub(col, add(margin, 2))
                    mstore(
                        stateM,
                        or(
                            and(
                                mload(
                                    add(
                                        add(
                                            GRID,
                                            shl(
                                                add(mload(LOG2_DIM_M), 5),
                                                and(sub(sub(row, add(margin, 2)), mul(shr(mload(LOG2_DIM_M), colN), shr(1, dim))), sub(dim, 1))
                                            )
                                        ),
                                        shl(5, and(colN, sub(dim, 1)))
                                    )
                                ),
                                0xfffffffffffffffff
                            ),
                            shl(
                                EXPAND_ROOT_OFFSET,
                                or(
                                    iszero(mload(OPEN_M)),
                                    and(
                                        and(
                                            and(gt(col, add(margin, 1)), lt(col, add(add(margin, 2), dim))),
                                            gt(shl(1, row), add(add(margin, 1), col))
                                        ),
                                        lt(shl(1, row), add(add(add(margin, 2), shl(1, dim)), col))
                                    )
                                )
                            )
                        )
                    )
                    if iszero(mload(OPEN_M)) {
                        mstore(stateM, or(and(mload(stateM), 0xfffffffffffff), shl(SELF_OFFSET, stateM)))
                    }
                }
                mstore(stateM, or(mload(stateM), shl(ROW_COL_OFFSET, or(shl(8, col), row))))
                for {
                    let angle
                } lt(angle, 6) {
                    angle := add(angle, 1)
                } {
                    let hexagonR0Plus := and(add(or(shl(8, col), row), shr(shl(4, angle), ANGLE_TO_HEXAGON_PLUS)), MASK_16)
                    mstore(
                        stateM,
                        or(
                            mload(stateM),
                            shl(
                                add(shl(4, angle), STATE_OFFSET),
                                add(add(unrolledGrid, mul(cols, shl(5, sub(and(hexagonR0Plus, MASK_8), 1)))), shl(5, sub(shr(8, hexagonR0Plus), 1)))
                            )
                        )
                    )
                }
            }
        }
    }

    function expand() internal pure {
        assembly {
            function expandDF() {
                for {

                } gt(mload(STACK_IDX_M), mload(STACK_M)) {

                } {
                    mstore(STACK_IDX_M, sub(mload(STACK_IDX_M), 0x3))
                    let top := shr(232, mload(mload(STACK_IDX_M)))
                    switch shr(20, top)
                    case 0 {
                        let colorStateM := and(shr(SELF_OFFSET, mload(top)), MASK_16)
                        for {
                            let angle
                        } lt(angle, 3) {
                            angle := add(angle, 1)
                        } {
                            if eq(and(shr(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(colorStateM)), MASK_4), mload(COLOR_M)) {
                                mstore(colorStateM, xor(mload(colorStateM), shl(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, angle)), top)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                            let stateR3M := and(shr(add(shl(4, add(angle, 3)), STATE_OFFSET), mload(top)), MASK_16)
                            let colorStateR3M := and(shr(SELF_OFFSET, mload(stateR3M)), MASK_16)
                            if and(
                                gt(colorStateR3M, 0),
                                eq(and(shr(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(colorStateR3M)), MASK_4), mload(COLOR_M))
                            ) {
                                mstore(colorStateR3M, xor(mload(colorStateR3M), shl(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, angle)), stateR3M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        mstore(top, xor(mload(top), HEXAGON_FULL_BOUNDARY_MASK))
                        mstore(
                            EDGE_COUNT_M,
                            sub(
                                add(
                                    mload(EDGE_COUNT_M),
                                    shl(1, and(shr(shl(2, and(shr(HEXAGON_EXPAND_OFFSET, mload(top)), MASK_6)), POPCOUNT_6), MASK_4))
                                ),
                                6
                            )
                        )
                        mstore(mload(CACHE_IDX_M), shl(232, top))
                        mstore(CACHE_IDX_M, add(mload(CACHE_IDX_M), 0x3))
                        mstore(mload(CACHE_IDX_M), shl(232, or(0x20000, top)))
                        mstore(CACHE_IDX_M, add(mload(CACHE_IDX_M), 0x3))
                        mstore(mload(CACHE_IDX_M), shl(232, or(0x40000, top)))
                        mstore(CACHE_IDX_M, add(mload(CACHE_IDX_M), 0x3))
                    }
                    case 1 {
                        let angle := and(shr(16, top), MASK_4)
                        let stateM := and(top, MASK_16)
                        let colorStateM := and(shr(SELF_OFFSET, mload(stateM)), MASK_16)
                        if eq(and(shr(HEXAGON_PAINT_OFFSET, mload(colorStateM)), MASK_4), mload(COLOR_M)) {
                            mstore(colorStateM, xor(mload(colorStateM), shl(HEXAGON_PAINT_OFFSET, mload(COLOR_M))))
                            mstore(mload(STACK_IDX_M), shl(232, stateM))
                            mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        }
                        let stateR0M := and(shr(add(shl(4, angle), STATE_OFFSET), mload(stateM)), MASK_16)
                        let colorStateR0M := and(shr(SELF_OFFSET, mload(stateR0M)), MASK_16)
                        if and(gt(colorStateR0M, 0), eq(and(shr(HEXAGON_PAINT_OFFSET, mload(colorStateR0M)), MASK_4), mload(COLOR_M))) {
                            mstore(colorStateR0M, xor(mload(colorStateR0M), shl(HEXAGON_PAINT_OFFSET, mload(COLOR_M))))
                            mstore(mload(STACK_IDX_M), shl(232, stateR0M))
                            mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        }
                        switch eq(angle, 2)
                        case 0 {
                            if eq(and(shr(add(shl(2, angle), TRIANGLE_PAINT_OFFSET), mload(colorStateM)), MASK_4), mload(COLOR_M)) {
                                mstore(colorStateM, xor(mload(colorStateM), shl(add(shl(2, angle), TRIANGLE_PAINT_OFFSET), mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(or(0x200000, shl(16, angle)), stateM)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        case 1 {
                            let state3M := and(shr(STATE_3_OFFSET, mload(stateM)), MASK_16)
                            let colorState3M := and(shr(SELF_OFFSET, mload(state3M)), MASK_16)
                            if and(gt(colorState3M, 0), eq(and(shr(TRIANGLE_PAINT_OFFSET, mload(colorState3M)), MASK_4), mload(COLOR_M))) {
                                mstore(colorState3M, xor(mload(colorState3M), shl(TRIANGLE_PAINT_OFFSET, mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(0x200000, state3M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        switch eq(angle, 0)
                        case 0 {
                            if eq(and(shr(add(shl(2, sub(angle, 1)), TRIANGLE_PAINT_OFFSET), mload(colorStateM)), MASK_4), mload(COLOR_M)) {
                                mstore(colorStateM, xor(mload(colorStateM), shl(add(shl(2, sub(angle, 1)), TRIANGLE_PAINT_OFFSET), mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(or(0x200000, shl(16, sub(angle, 1))), stateM)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        case 1 {
                            let state5M := and(shr(STATE_5_OFFSET, mload(stateM)), MASK_16)
                            let colorState5M := and(shr(SELF_OFFSET, mload(state5M)), MASK_16)
                            if and(gt(colorState5M, 0), eq(and(shr(TRIANGLE_1_PAINT_OFFSET, mload(colorState5M)), MASK_4), mload(COLOR_M))) {
                                mstore(colorState5M, xor(mload(colorState5M), shl(TRIANGLE_1_PAINT_OFFSET, mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(0x210000, state5M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        let angle3 := add(angle, 3)
                        mstore(stateM, xor(mload(stateM), shl(angle, SQUARE_HALF_BOUNDARY_MASK)))
                        mstore(stateR0M, xor(mload(stateR0M), shl(angle3, SQUARE_HALF_BOUNDARY_MASK)))
                        mstore(
                            EDGE_COUNT_M,
                            sub(
                                add(
                                    mload(EDGE_COUNT_M),
                                    shl(
                                        1,
                                        add(
                                            add(
                                                add(
                                                    and(shr(add(angle, HEXAGON_EXPAND_OFFSET), mload(stateM)), 1),
                                                    and(shr(add(angle, HEXAGON_PERP_EXPAND_OFFSET), mload(stateM)), 1)
                                                ),
                                                and(shr(add(angle3, HEXAGON_EXPAND_OFFSET), mload(stateR0M)), 1)
                                            ),
                                            and(shr(add(angle3, HEXAGON_PERP_EXPAND_OFFSET), mload(stateR0M)), 1)
                                        )
                                    )
                                ),
                                4
                            )
                        )
                        mstore(mload(CACHE_IDX_M), shl(232, or(shl(16, angle), stateM)))
                        mstore(CACHE_IDX_M, add(mload(CACHE_IDX_M), 0x3))
                        mstore(mload(CACHE_IDX_M), shl(232, or(shl(16, angle3), stateR0M)))
                        mstore(CACHE_IDX_M, add(mload(CACHE_IDX_M), 0x3))
                    }
                    case 2 {
                        let angle := and(shr(16, top), MASK_4)
                        let stateM := and(top, MASK_16)
                        let stateR0M := and(shr(add(shl(4, angle), STATE_OFFSET), mload(stateM)), MASK_16)
                        let stateR1M := and(shr(add(shl(4, add(angle, 1)), STATE_OFFSET), mload(stateM)), MASK_16)
                        let colorStateM := and(shr(SELF_OFFSET, mload(stateM)), MASK_16)
                        if eq(and(shr(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(colorStateM)), MASK_4), mload(COLOR_M)) {
                            mstore(colorStateM, xor(mload(colorStateM), shl(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                            mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, angle)), stateM)))
                            mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        }
                        if eq(and(shr(add(shl(2, add(angle, 1)), SQUARE_PAINT_OFFSET), mload(colorStateM)), MASK_4), mload(COLOR_M)) {
                            mstore(colorStateM, xor(mload(colorStateM), shl(add(shl(2, add(angle, 1)), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                            mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, add(angle, 1))), stateM)))
                            mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        }
                        switch angle
                        case 0 {
                            let colorStateR0M := and(shr(SELF_OFFSET, mload(stateR0M)), MASK_16)
                            if and(gt(colorStateR0M, 0), eq(and(shr(SQUARE_2_PAINT_OFFSET, mload(colorStateR0M)), MASK_4), mload(COLOR_M))) {
                                mstore(colorStateR0M, xor(mload(colorStateR0M), shl(SQUARE_2_PAINT_OFFSET, mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(0x120000, stateR0M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        case 1 {
                            let colorStateR1M := and(shr(SELF_OFFSET, mload(stateR1M)), MASK_16)
                            if and(gt(colorStateR1M, 0), eq(and(shr(SQUARE_PAINT_OFFSET, mload(colorStateR1M)), MASK_4), mload(COLOR_M))) {
                                mstore(colorStateR1M, xor(mload(colorStateR1M), shl(SQUARE_PAINT_OFFSET, mload(COLOR_M))))
                                mstore(mload(STACK_IDX_M), shl(232, or(0x100000, stateR1M)))
                                mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                            }
                        }
                        let angle2 := add(angle, 2)
                        let angle4 := add(angle, 4)
                        mstore(stateM, xor(mload(stateM), shl(angle, HEXAGON_PERP_EXPAND_MASK)))
                        mstore(stateR0M, xor(mload(stateR0M), shl(angle2, HEXAGON_PERP_EXPAND_MASK)))
                        mstore(stateR1M, xor(mload(stateR1M), shl(angle4, HEXAGON_PERP_EXPAND_MASK)))
                        mstore(
                            EDGE_COUNT_M,
                            sub(
                                add(
                                    mload(EDGE_COUNT_M),
                                    shl(
                                        1,
                                        add(
                                            add(
                                                and(shr(add(angle, HEXAGON_PERP_EXPAND_OFFSET), mload(stateM)), 1),
                                                and(shr(add(angle2, HEXAGON_PERP_EXPAND_OFFSET), mload(stateR0M)), 1)
                                            ),
                                            and(shr(add(angle4, HEXAGON_PERP_EXPAND_OFFSET), mload(stateR1M)), 1)
                                        )
                                    )
                                ),
                                3
                            )
                        )
                        mstore(mload(CACHE_IDX_M), shl(232, or(shl(16, angle), stateM)))
                        mstore(CACHE_IDX_M, add(mload(CACHE_IDX_M), 0x3))
                        mstore(mload(CACHE_IDX_M), shl(232, or(shl(16, angle2), stateR0M)))
                        mstore(CACHE_IDX_M, add(mload(CACHE_IDX_M), 0x3))
                    }
                }
            }
            function writeBoundary() {
                let outputIdx := mload(OUTPUT_IDX_M)
                mstore(outputIdx, '<path d="')
                outputIdx := add(outputIdx, 0x9)
                let cacheEnd := mload(CACHE_IDX_M)
                for {
                    let cacheIdx := mload(CACHE_M)
                } lt(cacheIdx, cacheEnd) {
                    cacheIdx := add(cacheIdx, 0x3)
                } {
                    let vertex := shr(232, mload(cacheIdx))
                    let stateM := and(vertex, MASK_16)
                    let angle := and(shr(16, vertex), MASK_4)
                    if iszero(and(mload(stateM), shl(HEXAGON_EXPAND_OFFSET, and(shr(mul(angle, 12), VERTEX_013_EXPAND_MASK), MASK_12)))) {
                        continue
                    }
                    mstore8(outputIdx, 0x4d)
                    outputIdx := add(outputIdx, 0x1)
                    let hexagon := and(shr(ROW_COL_OFFSET, mload(stateM)), MASK_16)
                    {
                        let vx := and(add(mul(sub(shr(8, hexagon), add(mload(MARGIN_M), 2)), 4732), shr(shl(5, angle), ANGLE_TO_VERTEX_X)), MASK_32)
                        if shr(31, vx) {
                            vx := and(add(not(vx), 1), MASK_32)
                            mstore8(outputIdx, 0x2d)
                            outputIdx := add(outputIdx, 0x1)
                        }
                        let length := 1
                        let a := vx
                        if gt(a, 9999) {
                            length := add(length, 4)
                            a := div(a, 10000)
                        }
                        if gt(a, 99) {
                            length := add(length, 2)
                            a := div(a, 100)
                        }
                        if gt(a, 9) {
                            length := add(length, 1)
                        }
                        let p := add(outputIdx, length)
                        for {

                        } gt(p, outputIdx) {

                        } {
                            p := sub(p, 0x1)
                            mstore8(p, add(mod(vx, 10), 48))
                            vx := div(vx, 10)
                        }
                        outputIdx := add(outputIdx, length)
                    }
                    mstore8(outputIdx, 0x20)
                    outputIdx := add(outputIdx, 0x1)
                    {
                        let vy := and(
                            add(
                                mul(sub(sub(shl(1, and(hexagon, MASK_8)), shr(8, hexagon)), add(mload(MARGIN_M), 2)), 2732),
                                shr(shl(5, angle), ANGLE_TO_VERTEX_Y)
                            ),
                            MASK_32
                        )
                        if shr(31, vy) {
                            vy := and(add(not(vy), 1), MASK_32)
                            mstore8(outputIdx, 0x2d)
                            outputIdx := add(outputIdx, 0x1)
                        }
                        let length := 1
                        let a := vy
                        if gt(a, 9999) {
                            length := add(length, 4)
                            a := div(a, 10000)
                        }
                        if gt(a, 99) {
                            length := add(length, 2)
                            a := div(a, 100)
                        }
                        if gt(a, 9) {
                            length := add(length, 1)
                        }
                        let p := add(outputIdx, length)
                        for {

                        } gt(p, outputIdx) {

                        } {
                            p := sub(p, 0x1)
                            mstore8(p, add(mod(vy, 10), 48))
                            vy := div(vy, 10)
                        }
                        outputIdx := add(outputIdx, length)
                    }
                    mstore8(outputIdx, 0x6c)
                    outputIdx := add(outputIdx, 0x1)
                    let edgeCount := mload(EDGE_COUNT_M)
                    for {

                    } 1 {

                    } {
                        if and(mload(stateM), shl(angle, HEXAGON_EXPAND_MASK)) {
                            let adjusted := shr(shl(5, angle), ANGLE_EDGE_TO_VECTOR_OFFSET)
                            let offset := and(adjusted, MASK_8)
                            mstore(outputIdx, mload(add(ANGLE_EDGE_TO_VECTOR, offset)))
                            outputIdx := add(outputIdx, sub(and(shr(8, adjusted), MASK_8), offset))
                            mstore(stateM, xor(mload(stateM), shl(angle, HEXAGON_EXPAND_MASK)))
                            angle := addmod(angle, 5, 6)
                            edgeCount := sub(edgeCount, 1)
                            continue
                        }
                        if and(mload(stateM), shl(angle, HEXAGON_PERP_EXPAND_MASK)) {
                            let adjusted := shr(add(shl(5, angle), 8), ANGLE_EDGE_TO_VECTOR_OFFSET)
                            let offset := and(adjusted, MASK_8)
                            mstore(outputIdx, mload(add(ANGLE_EDGE_TO_VECTOR, offset)))
                            outputIdx := add(outputIdx, sub(and(shr(8, adjusted), MASK_8), offset))
                            mstore(stateM, xor(mload(stateM), shl(angle, HEXAGON_PERP_EXPAND_MASK)))
                            stateM := and(shr(add(shl(4, angle), STATE_OFFSET), mload(stateM)), MASK_16)
                            angle := addmod(angle, 2, 6)
                            edgeCount := sub(edgeCount, 1)
                            continue
                        }
                        let stateR1M := and(shr(add(shl(4, addmod(angle, 1, 6)), STATE_OFFSET), mload(stateM)), MASK_16)
                        if and(mload(stateR1M), shl(addmod(angle, 4, 6), HEXAGON_PERP_EXPAND_MASK)) {
                            let adjusted := shr(add(shl(5, angle), 16), ANGLE_EDGE_TO_VECTOR_OFFSET)
                            let offset := and(adjusted, MASK_8)
                            mstore(outputIdx, mload(add(ANGLE_EDGE_TO_VECTOR, offset)))
                            outputIdx := add(outputIdx, sub(and(shr(8, adjusted), MASK_8), offset))
                            stateM := stateR1M
                            angle := addmod(angle, 4, 6)
                            mstore(stateM, xor(mload(stateM), shl(angle, HEXAGON_PERP_EXPAND_MASK)))
                            edgeCount := sub(edgeCount, 1)
                            continue
                        }
                        if and(mload(stateM), shl(addmod(angle, 1, 6), HEXAGON_EXPAND_MASK)) {
                            let adjusted := shr(add(shl(5, angle), 24), ANGLE_EDGE_TO_VECTOR_OFFSET)
                            let offset := and(adjusted, MASK_8)
                            mstore(outputIdx, mload(add(ANGLE_EDGE_TO_VECTOR, offset)))
                            outputIdx := add(outputIdx, sub(and(shr(8, adjusted), MASK_8), offset))
                            angle := addmod(angle, 1, 6)
                            mstore(stateM, xor(mload(stateM), shl(angle, HEXAGON_EXPAND_MASK)))
                            edgeCount := sub(edgeCount, 1)
                            continue
                        }
                        break
                    }
                    mstore(EDGE_COUNT_M, edgeCount)
                    if iszero(edgeCount) {
                        break
                    }
                }
                mstore(outputIdx, '" fill="#')
                outputIdx := add(outputIdx, 0x9)
                mstore(outputIdx, mload(add(mload(PALETTE_M), mul(sub(mload(COLOR_M), 1), 0x6))))
                outputIdx := add(outputIdx, 0x6)
                mstore(outputIdx, '"/>')
                outputIdx := add(outputIdx, 0x3)
                mstore(OUTPUT_IDX_M, outputIdx)
            }
            if mload(OPEN_M) {
                mstore(RANDOM_SOURCE, keccak256(RANDOM_SOURCE, 0x20))
                mstore(SHIFT_M, 0)
                let sampleCount := mul(mload(DIM_M), mload(DIM_M))
                for {
                    let i
                } lt(i, sampleCount) {
                    i := add(i, 1)
                } {
                    let shift := mload(SHIFT_M)
                    let stateM := add(
                        mload(UNROLLED_GRID_M),
                        shl(
                            5,
                            shr(
                                16,
                                mul(mul(mload(UNROLLED_GRID_ROWS_M), mload(UNROLLED_GRID_COLS_M)), and(shr(shift, mload(RANDOM_SOURCE)), MASK_16))
                            )
                        )
                    )
                    shift := add(shift, 16)
                    if eq(shift, 256) {
                        mstore(RANDOM_SOURCE, keccak256(RANDOM_SOURCE, 0x20))
                        shift := 0
                    }
                    mstore(SHIFT_M, shift)
                    let colorStateM := and(shr(SELF_OFFSET, mload(stateM)), MASK_16)
                    if and(gt(colorStateM, 0), gt(and(mload(colorStateM), HEXAGON_PAINT_MASK), 0)) {
                        mstore(COLOR_M, and(shr(HEXAGON_PAINT_OFFSET, mload(colorStateM)), MASK_4))
                        mstore(colorStateM, xor(mload(colorStateM), shl(HEXAGON_PAINT_OFFSET, mload(COLOR_M))))
                        mstore(mload(STACK_IDX_M), shl(232, stateM))
                        mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        mstore(CACHE_IDX_M, mload(CACHE_M))
                        expandDF()
                        writeBoundary()
                    }
                }
            }
            let unrolledGridEnd := mload(STACK_M)
            for {
                let stateM := mload(UNROLLED_GRID_M)
            } lt(stateM, unrolledGridEnd) {
                stateM := add(stateM, 0x20)
            } {
                if iszero(and(mload(stateM), shl(EXPAND_ROOT_OFFSET, 1))) {
                    continue
                }
                let colorStateM := and(shr(SELF_OFFSET, mload(stateM)), MASK_16)
                if and(mload(colorStateM), HEXAGON_PAINT_MASK) {
                    mstore(COLOR_M, and(shr(HEXAGON_PAINT_OFFSET, mload(colorStateM)), MASK_4))
                    mstore(colorStateM, xor(mload(colorStateM), shl(HEXAGON_PAINT_OFFSET, mload(COLOR_M))))
                    mstore(mload(STACK_IDX_M), shl(232, stateM))
                    mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                    mstore(CACHE_IDX_M, mload(CACHE_M))
                    expandDF()
                    writeBoundary()
                }
                for {
                    let angle
                } lt(angle, 3) {
                    angle := add(angle, 1)
                } {
                    if and(mload(colorStateM), shl(shl(2, angle), SQUARE_PAINT_MASK)) {
                        mstore(COLOR_M, and(shr(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(colorStateM)), MASK_4))
                        mstore(colorStateM, xor(mload(colorStateM), shl(add(shl(2, angle), SQUARE_PAINT_OFFSET), mload(COLOR_M))))
                        mstore(mload(STACK_IDX_M), shl(232, or(or(0x100000, shl(16, angle)), stateM)))
                        mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        mstore(CACHE_IDX_M, mload(CACHE_M))
                        expandDF()
                        writeBoundary()
                    }
                    if and(lt(angle, 2), gt(and(mload(colorStateM), shl(shl(2, angle), TRIANGLE_PAINT_MASK)), 0)) {
                        mstore(COLOR_M, and(shr(add(shl(2, angle), TRIANGLE_PAINT_OFFSET), mload(colorStateM)), MASK_4))
                        mstore(colorStateM, xor(mload(colorStateM), shl(add(shl(2, angle), TRIANGLE_PAINT_OFFSET), mload(COLOR_M))))
                        mstore(mload(STACK_IDX_M), shl(232, or(or(0x200000, shl(16, angle)), stateM)))
                        mstore(STACK_IDX_M, add(mload(STACK_IDX_M), 0x3))
                        mstore(CACHE_IDX_M, mload(CACHE_M))
                        expandDF()
                        writeBoundary()
                    }
                }
            }
        }
    }

    function writeDecimalLookup() internal pure {
        assembly {
            // Assume z is signed 32-bit and |z| < 10 ** 8
            function writeDecimal(z, decimalM, decimalLengthM) {
                let outputIdx := mload(OUTPUT_IDX_M)
                mstore(decimalM, outputIdx)
                if shr(31, z) {
                    z := and(add(not(z), 1), MASK_32)
                    mstore8(outputIdx, 0x2d)
                    outputIdx := add(outputIdx, 0x1)
                }
                let length := 1
                let a := z
                if gt(a, 9999) {
                    length := add(length, 4)
                    a := div(a, 10000)
                }
                if gt(a, 99) {
                    length := add(length, 2)
                    a := div(a, 100)
                }
                if gt(a, 9) {
                    length := add(length, 1)
                }
                let p := add(outputIdx, length)
                for {

                } gt(p, outputIdx) {

                } {
                    p := sub(p, 0x1)
                    mstore8(p, add(mod(z, 10), 48))
                    z := div(z, 10)
                }
                outputIdx := add(outputIdx, length)
                mstore(decimalLengthM, sub(outputIdx, mload(decimalM)))
                mstore(OUTPUT_IDX_M, outputIdx)
            }
            writeDecimal(mul(sub(0, add(mload(MARGIN_M), 2)), 4732), OPEN_VIEW_BOX_X_DECIMAL_M, OPEN_VIEW_BOX_X_DECIMAL_LENGTH_M)
            writeDecimal(mul(sub(0, add(mload(MARGIN_M), 4)), 2732), OPEN_VIEW_BOX_Y_DECIMAL_M, OPEN_VIEW_BOX_Y_DECIMAL_LENGTH_M)
            writeDecimal(
                mul(add(add(mload(DIM_M), 3), shl(1, mload(MARGIN_M))), 4732),
                OPEN_VIEW_BOX_WIDTH_DECIMAL_M,
                OPEN_VIEW_BOX_WIDTH_DECIMAL_LENGTH_M
            )
            writeDecimal(mul(add(add(mload(DIM_M), 3), mload(MARGIN_M)), 5464), OPEN_VIEW_BOX_HEIGHT_DECIMAL_M, OPEN_VIEW_BOX_HEIGHT_DECIMAL_LENGTH_M)
            writeDecimal(mul(mload(DIM_M), 4732), DOMAIN_WIDTH_DECIMAL_M, DOMAIN_WIDTH_DECIMAL_LENGTH_M)
            writeDecimal(mul(mload(DIM_M), 5464), DOMAIN_HEIGHT_DECIMAL_M, DOMAIN_HEIGHT_DECIMAL_LENGTH_M)
            writeDecimal(mload(TOKEN_ID_M), TOKEN_ID_DECIMAL_M, TOKEN_ID_DECIMAL_LENGTH_M)
            writeDecimal(mload(DIM_M), DIM_DECIMAL_M, DIM_DECIMAL_LENGTH_M)
            writeDecimal(mload(PALETTE_IDX_M), PALETTE_IDX_DECIMAL_M, PALETTE_IDX_DECIMAL_LENGTH_M)
        }
    }

    function writePreExpand() internal pure {
        assembly {
            let outputIdx := mload(OUTPUT_IDX_M)
            // '<svg xmlns="http://www.w3.org/2000/svg" viewBox="'
            mstore(outputIdx, mload(mload(SVG_STRING_LOOKUP_M)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x20)))
            outputIdx := add(outputIdx, 0x11)
            switch mload(OPEN_M)
            case 0 {
                // '-2732 -2732 '
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xfa)))
                outputIdx := add(outputIdx, 0xc)
                mstore(outputIdx, mload(mload(DOMAIN_WIDTH_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(DOMAIN_WIDTH_DECIMAL_LENGTH_M))
                mstore8(outputIdx, 0x20)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(DOMAIN_HEIGHT_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(DOMAIN_HEIGHT_DECIMAL_LENGTH_M))
                // '">'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x31)))
                outputIdx := add(outputIdx, 0x2)
            }
            case 1 {
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_X_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_X_DECIMAL_LENGTH_M))
                mstore8(outputIdx, 0x20)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_Y_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_Y_DECIMAL_LENGTH_M))
                mstore8(outputIdx, 0x20)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_WIDTH_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_WIDTH_DECIMAL_LENGTH_M))
                mstore8(outputIdx, 0x20)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_HEIGHT_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_HEIGHT_DECIMAL_LENGTH_M))
                // '"><rect x="'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x31)))
                outputIdx := add(outputIdx, 0xb)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_X_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_X_DECIMAL_LENGTH_M))
                // '" y="'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x41)))
                outputIdx := add(outputIdx, 0x5)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_Y_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_Y_DECIMAL_LENGTH_M))
                // '" width="'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x4b)))
                outputIdx := add(outputIdx, 0x9)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_WIDTH_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_WIDTH_DECIMAL_LENGTH_M))
                // '" height="'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x54)))
                outputIdx := add(outputIdx, 0xa)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_HEIGHT_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_HEIGHT_DECIMAL_LENGTH_M))
                // '" fill="white"/>'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x5e)))
                outputIdx := add(outputIdx, 0x10)
                // '<rect x="-2732" y="-2732" width='
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x33)))
                outputIdx := add(outputIdx, 0x20)
                mstore8(outputIdx, 0x22)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(DOMAIN_WIDTH_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(DOMAIN_WIDTH_DECIMAL_LENGTH_M))
                // '" height="'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x54)))
                outputIdx := add(outputIdx, 0xa)
                mstore(outputIdx, mload(mload(DOMAIN_HEIGHT_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(DOMAIN_HEIGHT_DECIMAL_LENGTH_M))
                // '" '
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x5e)))
                outputIdx := add(outputIdx, 0x2)
                // 'stroke="black" stroke-width="100'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x71)))
                outputIdx := add(outputIdx, 0x20)
                // '" fill="white"/>'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x5e)))
                outputIdx := add(outputIdx, 0x10)
            }
            // '<g stroke="black" stroke-width="100" stroke-linejoin="round" stroke-linecap="round" fill-rule="evenodd">'
            mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x6e)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x8e)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xae)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xce)))
            outputIdx := add(outputIdx, 0x8)
            mstore(OUTPUT_IDX_M, outputIdx)
        }
    }

    function writePostExpand() internal pure {
        assembly {
            let outputIdx := mload(OUTPUT_IDX_M)
            // '</g>'
            mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xeb)))
            outputIdx := add(outputIdx, 0x4)
            if iszero(mload(OPEN_M)) {
                // '<path d="'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xd6)))
                outputIdx := add(outputIdx, 0x9)
                mstore8(outputIdx, 0x4d)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_X_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_X_DECIMAL_LENGTH_M))
                mstore8(outputIdx, 0x20)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_Y_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_Y_DECIMAL_LENGTH_M))
                mstore8(outputIdx, 0x6c)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_WIDTH_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_WIDTH_DECIMAL_LENGTH_M))
                // ' 0 0 '
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xf5)))
                outputIdx := add(outputIdx, 0x5)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_HEIGHT_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_HEIGHT_DECIMAL_LENGTH_M))
                // ' -'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xf9)))
                outputIdx := add(outputIdx, 0x2)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_WIDTH_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_WIDTH_DECIMAL_LENGTH_M))
                // ' 0 0 -'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xf5)))
                outputIdx := add(outputIdx, 0x6)
                mstore(outputIdx, mload(mload(OPEN_VIEW_BOX_HEIGHT_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(OPEN_VIEW_BOX_HEIGHT_DECIMAL_LENGTH_M))
                // 'M-2732 -2732l0 '
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x106)))
                outputIdx := add(outputIdx, 0xf)
                mstore(outputIdx, mload(mload(DOMAIN_HEIGHT_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(DOMAIN_HEIGHT_DECIMAL_LENGTH_M))
                mstore8(outputIdx, 0x20)
                outputIdx := add(outputIdx, 0x1)
                mstore(outputIdx, mload(mload(DOMAIN_WIDTH_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(DOMAIN_WIDTH_DECIMAL_LENGTH_M))
                // ' 0 0 -'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xf5)))
                outputIdx := add(outputIdx, 0x6)
                mstore(outputIdx, mload(mload(DOMAIN_HEIGHT_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(DOMAIN_HEIGHT_DECIMAL_LENGTH_M))
                // ' -'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xf9)))
                outputIdx := add(outputIdx, 0x2)
                mstore(outputIdx, mload(mload(DOMAIN_WIDTH_DECIMAL_M)))
                outputIdx := add(outputIdx, mload(DOMAIN_WIDTH_DECIMAL_LENGTH_M))
                // ' 0'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xf5)))
                outputIdx := add(outputIdx, 0x2)
                // '" fill="white"/>'
                mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0x5e)))
                outputIdx := add(outputIdx, 0x10)
            }
            // '</svg>'
            mstore(outputIdx, mload(add(mload(SVG_STRING_LOOKUP_M), 0xef)))
            outputIdx := add(outputIdx, 0x6)
            mstore(outputIdx, 0)
            mstore(OUTPUT_IDX_M, outputIdx)
            mstore(SVG_START_M, mload(OUTPUT_M))
            mstore(SVG_END_M, outputIdx)
        }
    }

    function writeJSON() internal pure {
        assembly {
            let outputIdx := mload(OUTPUT_IDX_M)
            // 'data:application/json,%7B%22name%22:%22Tiling%20'
            mstore(outputIdx, mload(mload(JSON_STRING_LOOKUP_M)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x20)))
            outputIdx := add(outputIdx, 0x10)
            mstore(outputIdx, mload(mload(TOKEN_ID_DECIMAL_M)))
            outputIdx := add(outputIdx, mload(TOKEN_ID_DECIMAL_LENGTH_M))
            // '%22,%22description%22:%22Hexamillennia%20is%20generated%20entirely%20on%20the%20EVM.%20Released%20under%20CC0.%22,%22attributes%22:%5B%7B%22trait_type%22:%22'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x30)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x50)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x70)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x90)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0xb0)))
            outputIdx := add(outputIdx, 0x1d)
            // 'Size'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x14a)))
            outputIdx := add(outputIdx, 0x4)
            // '%22,%22value%22:%22'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0xcd)))
            outputIdx := add(outputIdx, 0x13)
            mstore(outputIdx, mload(mload(DIM_DECIMAL_M)))
            outputIdx := add(outputIdx, mload(DIM_DECIMAL_LENGTH_M))
            // '%22%7D,%7B%22trait_type%22:%22'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0xe0)))
            outputIdx := add(outputIdx, 0x1e)
            // 'Form'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x14e)))
            outputIdx := add(outputIdx, 0x4)
            // '%22,%22value%22:%22'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0xcd)))
            outputIdx := add(outputIdx, 0x13)
            switch mload(OPEN_M)
            case 0 {
                // 'Closed'
                mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x15e)))
                outputIdx := add(outputIdx, 0x6)
            }
            case 1 {
                // 'Open'
                mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x164)))
                outputIdx := add(outputIdx, 0x4)
            }
            // '%22%7D,%7B%22trait_type%22:%22'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0xe0)))
            outputIdx := add(outputIdx, 0x1e)
            // 'Steps'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x152)))
            outputIdx := add(outputIdx, 0x5)
            // '%22,%22value%22:%22'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0xcd)))
            outputIdx := add(outputIdx, 0x13)
            switch mload(STEPS_IDX_M)
            case 0 {
                // 'Low'
                mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x168)))
                outputIdx := add(outputIdx, 0x3)
            }
            case 1 {
                // 'Medium'
                mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x16b)))
                outputIdx := add(outputIdx, 0x6)
            }
            case 2 {
                // 'High'
                mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x171)))
                outputIdx := add(outputIdx, 0x4)
            }
            // '%22%7D,%7B%22trait_type%22:%22'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0xe0)))
            outputIdx := add(outputIdx, 0x1e)
            // 'Palette'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x157)))
            outputIdx := add(outputIdx, 0x7)
            // '%22,%22value%22:%22'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0xcd)))
            outputIdx := add(outputIdx, 0x13)
            mstore(outputIdx, mload(mload(PALETTE_IDX_DECIMAL_M)))
            outputIdx := add(outputIdx, mload(PALETTE_IDX_DECIMAL_LENGTH_M))
            // '%22%7D%5D,%22image%22:%22data:image/svg+xml;base64,'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x111)))
            outputIdx := add(outputIdx, 0x20)
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x131)))
            outputIdx := add(outputIdx, 0x13)
            // Base64 encode
            //
            // Adapted from https://github.com/OpenZeppelin/openzeppelin-contracts/blob/master/contracts/utils/Base64.sol
            let end := sub(mload(SVG_END_M), 0x20)
            for {
                let svgIdx := sub(mload(SVG_START_M), 0x20)
            } lt(svgIdx, end) {

            } {
                svgIdx := add(svgIdx, 0x3)
                let input := mload(svgIdx)
                mstore8(outputIdx, mload(add(BASE64, and(shr(18, input), MASK_6))))
                mstore8(add(outputIdx, 0x1), mload(add(BASE64, and(shr(12, input), MASK_6))))
                mstore8(add(outputIdx, 0x2), mload(add(BASE64, and(shr(6, input), MASK_6))))
                mstore8(add(outputIdx, 0x3), mload(add(BASE64, and(input, MASK_6))))
                outputIdx := add(outputIdx, 0x4)
            }
            switch mod(sub(mload(SVG_END_M), mload(SVG_START_M)), 3)
            case 1 {
                mstore8(sub(outputIdx, 0x1), 0x3d)
                mstore8(sub(outputIdx, 0x2), 0x3d)
            }
            case 2 {
                mstore8(sub(outputIdx, 0x1), 0x3d)
            }
            // '%22%7D'
            mstore(outputIdx, mload(add(mload(JSON_STRING_LOOKUP_M), 0x144)))
            outputIdx := add(outputIdx, 0x6)
            mstore(outputIdx, 0)
            mstore(OUTPUT_IDX_M, outputIdx)
        }
    }

    function resetOutput() internal pure {
        assembly {
            mstore(OUTPUT_M, add(mload(OUTPUT_IDX_M), 0x40))
            mstore(OUTPUT_IDX_M, mload(OUTPUT_M))
        }
    }

    function returnOutput() internal pure {
        assembly {
            let output := mload(OUTPUT_M)
            let length := sub(mload(OUTPUT_IDX_M), output)
            mstore(sub(output, 0x40), 0x20)
            mstore(sub(output, 0x20), length)
            return(sub(output, 0x40), add(shl(5, shr(5, add(length, 31))), 0x40))
        }
    }
}